
UDTFs: The Unsung DB2 Function 
 
The acronym UDTF stands for user-defined table function, but it could very well stand 
for unsung DB2 table function when compared with its counterpart, the user-defined 
scalar function (UDF). From this article, you will gain a better appreciation of how you 
can use table functions to encapsulate complex SQL processing and more importantly, 
how you can use UDTFs from SQL interfaces to access non-relational data sources such 
as S/36 files and stream files residing in IFS. 

In case you missed it, DB2 for i5/OS shipped support for UDTFs in V5R2. The 
UDTF enhancement complemented the UDF capability added way back in V4R4 of 
OS/400. A UDF classified as scalar means that the function can return only a single 
value (or output parameter). Many programmers have successfully used this function type 
as they created their own functions for manipulating data (e.g., converting packed date 
value to SQL date format) or reusing business calculations in existing RPG programs. 

Although having a UDF send back a single value worked well in these cases, there 
were other problems that required returning multiple values. That’s where UDTFs come 
into play with their ability to return multiple values in the form of a table (or result set). 
Another difference between the two function types is that because a UDTF can return 
multiple values, it can be invoked only on the FROM clause of a SELECT statement. 
 

Figure 1A: SQL request using both types of UDFs 
 
SELECT emp_id, emp_name, Ed_Degree(educ_level)  
    FROM TABLE(deptemployees('503')) myudtf 
 
The example in Figure 1A uses both types of functions in the same SQL request to 

demonstrate the differences between the two. This SELECT statement (1) invokes the 
DeptEmployees UDTF to return a list of employees from a specific department and (2) 
invokes the Ed_Degree UDF to output a text description of an employee’s education 
level, resulting in the final result shown in Figure 1B. 

 
Figure 1B: Result of sample SQL request 

 
 

Figure 2 shows the source code for the UDF. You’ll see that it returns a single output 
value — a variable-length character string with a user-friendly text description of a 
numeric education level. The Ed_Degree UDF is invoked by DB2 each time that a row is 
selected to return a single text description value back to the invoker. 
 



Figure 2: Source code for the Ed_Degree UDF 
 

CREATE FUNCTION ed_degree(i_level SMALLINT) 
  RETURNS VARCHAR(30) 
  LANGUAGE SQL 
  DETERMINISTIC 
  NOT FENCED 
BEGIN 
  CASE i_level  
    WHEN 12 THEN RETURN 'High School'; 
    WHEN 13 THEN RETURN 'College'; 
    WHEN 14 THEN RETURN 'Masters'; 
    WHEN 15 THEN RETURN 'Doctorate'; 
    ELSE         RETURN 'Unknown'; 
  END CASE; 
END; 
 
In contrast, the RETURNS clause found in the UDTF source code in Figure 3 clearly 

enables the function to return four different output values. 
 
Figure 3: Source code for DeptEmployees UDTF 
 
CREATE FUNCTION DeptEmployees (i_deptno VARCHAR(3)) 
  RETURNS TABLE (emp_id CHAR(6), emp_name VARCHAR(35), 
                 educ_level SMALLINT, dept_name CHAR(20)) 
  LANGUAGE SQL  
  DETERMINISTIC 
  NOT FENCED 
  NO EXTERNAL ACTION  
  DISALLOW PARALLEL 
  CARDINALITY 10 
  RETURN 
      SELECT empno, lastname||’, ‘||firstnme, edlevel, deptname 
          FROM employee e, dept d 
          WHERE e.workdept=d.deptno AND 
                        e.workdept = i_deptno; 
 
Furthermore, if there is more than one employee in the department '503', then the 

DeptEmployees UDTF will return these four output values for each employee. The table 
(or result set) returned by DeptEmployees will contain a row for each employee in 
department '503' with the column names for the result table being emp_id, emp_name, 
educ_level, and dept_name. 

So, the output of a UDTF can really be considered both wider and deeper than a 
UDF. A table function can output a maximum of 123 values. 

This is a pretty simple example of a UDTF, but it demonstrates how you can use a 
UDTF to hide or encapsulate some complex SQL processing inside the function 
definition. This is similar to the capabilities provided by an SQL view, but a table 
function has the advantage of being able to receive input parameters (up to 90) and 
perform inserts, updates, and deletes before returning a result set. 

Because SQL UDTFs can use the SQL Procedural Language (PL), a UDTF can also 
perform conditional processing as part of its logic to generate the return table. As a more 
advanced UDTF example (Figure 4) demonstrates, you can use the SQL PL to make the 



department search case-insensitive and ensure that the specified education level 
parameter value is valid. In addition, the Adv_DeptEmployees UDTF inserts a row into 
the table. 

 
Figure 4: A more advanced UDTF example 
 
CREATE FUNCTION Adv_DeptEmployees(i_deptno CHAR(3),  
                                  i_level SMALLINT) 
  RETURNS TABLE (emp_id CHAR(6), emp_name VARCHAR(35), 
                 educ_level SMALLINT, dept_name CHAR(20)) 
  LANGUAGE SQL 
  DETERMINISTIC 
  NOT FENCED 
  EXTERNAL ACTION 
  DISALLOW PARALLEL 
  CARDINALITY 10 
BEGIN 
 DECLARE AllCaps_DeptNum CHAR(3); 
 
 SET AllCaps_DeptNum=UPPER(i_deptno); 
 
 IF (i_level > 15 OR i_level<12) THEN 
   SIGNAL ‘75002’ SET MESSAGE_TEXT=’Invalid education level’; 
 
 INSERT INTO audit_employee_access VALUES(USER, CURRENT TIMESTAMP); 
 
 RETURN   
      SELECT empno, lastname||', '||firstnme, edlevel, deptname 
          FROM employee e , dept d  
      WHERE e.workdept=AllCaps_DeptNum and e.edlevel=i_level; 
END; 
 
Like a UDF, a UDTF can be written either in SQL or as an external function using 

any System i high-level programming language such as RPG, Cobol, or Java. 
You can actually join a UDTF to another table as part of its usage, which makes 

sense because a UDTF outputs a result table. Figure 5 shows an expanded version of the 
previous SELECT statement to include the project names for an employee to demonstrate 
this capability. 

 
Figure 5: Expanded version of previous SELECT statement 
 
SELECT emp_id, emp_name, projname, Ed_Degree(educ_level)  
    FROM TABLE(deptemployees('503')) myudtf JOIN project p 
     ON myudtf.emp_id = p.respemp  
WHERE  p.prendate < CURRENT DATE 
 
The table functions from previous examples contain several options (e.g., 

DETERMINISTIC) that we have not yet discussed. The syntax diagram in Figure 6 
details a complete list of the table functions attributes. These options are documented in 
the SQL Reference manual, so let’s concentrate on some of the more interesting options. 

 



Figure 6: SQL UDTF Syntax Diagram 
 

 
 
 
The FENCED or NOT FENCED option is the first attribute to highlight. FENCED is 

the default attribute and is supported primarily because other DB2 products need the 
option to control their memory usage during the execution of a UDF. This aspect does not 
apply to DB2 for i5/OS because the database engine is part of the operating system. Thus, 
DB2’s memory usage cannot be separated from the memory used by i5/OS. 

The FENCED attribute causes the UDF to perform slower because it causes DB2 to 
perform the UDF call in a different thread. The NOT FENCED attribute allows DB2 to 
execute the UDF call within the same thread as the invoking SQL statement. Currently, 
the NOT FENCED attribute is not supported for UDTF invocations, but IBM will add 
this support in a future release. So despite the lack of UDTFs today, you can use the NOT 
FENCED attribute now, and you will automatically benefit when IBM adds support in a 
future release. 

The DETERMINISTIC option is another option that can improve performance, but 
not until a future release. The DETERMINISTIC setting says that a UDF will always 
return the same result value when called with the exact same input parameters. For 
example, the Ed_Degree function is deterministic because each time that it’s invoked 
with an input parameter value of 13, it will output the 'College' text string. 

The main performance benefit of a DETERMINISTIC function is that it allows DB2 
to cache the input parameter values and associated return value for a UDF call. On 
successive calls to that UDF with the same input parameters, DB2 can just return the 
cached returned value without ever invoking the UDF. Performance is gained by 
eliminating the overhead involved with calling the UDF. Remember that when DB2 
invokes a UDF, it’s equivalent to the performance of executing an unbound, external 
program call. Avoiding the function call can produce noticeable performance gains. 

The EXTERNAL ACTION option also must be considered because it can cause a 
deterministic function to always be invoked as opposed to using cached results. When a 
table function performs an action other than returning output values (such as inserting a 
row into a table or putting an entry on a data queue), the EXTERNAL ACTION option 



must be specified to guarantee that the function is called on successive invocations. In 
Figure 4, the EXTERNAL ACTION option had to be added to the advanced version of 
the table function because an audit record is written to table before returning the output. 

Like the DETERMINISTIC and NOT FENCED attributes, the Cardinality option 
can also affect the performance of a table function. The cardinality provides the query 
optimizer an estimate of the number of rows that will be returned by the table function. 
This estimate of the number of rows will help the query optimizer determine the most 
efficient join order in cases such as the join example in Figure 5. 

If the table function returns more or less rows than the specified cardinality, a 
runtime error will not occur because this cardinality value is used only during query 
optimization. 
 
External Table Functions 

Now, that you have an understanding of SQL table functions. Let’s move onto the 
external table functions. An external UDTF means that you can use your favorite System 
i high-level programming language to implement the table function. External table 
functions provide the overlooked ability mentioned at the beginning of the article: the 
ability to return nonrelational data (e.g., IFS stream files) in a relational format to SQL 
interfaces such as JDBC. 

Let’s review an example of an external UDTF that accesses and returns data from 
S/36 files. The program object used for an external UDTF is not required to use SQL. 
This capability is what allows it to access data sources outside of DB2 and transform 
them into relational data. The only SQL that is required for an external UDTF is the 
CREATE FUNCTION statement to register an i5/OS (or OS/400) program object as a 
table function. 

The CREATE FUNCTION statement in Figure 7 registers an ILE RPG program, 
EMPDEPUDTF, as a table function. This table function returns information for a 
specified employee if an employee value is specified. The underlying RPG program that 
will be reviewed later on is designed to return all of the employees when the employee 
number input value is a blank string. 

 



Figure 7: CREATE FUNCTION statement 
 
CREATE FUNCTION EmployeesDept(i_empno VARCHAR(6)) 
      RETURNS TABLE 
      ( empno      varchar(6), 
        firstname  varchar(14), 
        lastname   varchar(17), 
        workdept   varchar(3), 
        departname varchar(38), 
        location   varchar(16), 
        salary     dec(9,2) 
      ) 
LANGUAGE RPGLE 
NO SQL 
EXTERNAL EMPDEPUDTF 
PARAMETER STYLE DB2SQL 
SCRATCHPAD 10 
NO FINAL CALL 
DISALLOW PARALLEL 
NO EXTERNAL ACTION 
DETERMINISTIC 
NOT FENCED 
CARDINALITY 10; 
 
Notice that the statement contains several new function attributes that we didn’t 

cover earlier. External UDTFs have a different set of attributes available when you’re 
registering the external program as a function (Figure 8). The EXTERNAL clause is used 
to simply identify the i5/OS program or service program object that is being registered as 
a table function. The NO SQL clause is also straightforward, as it specifies that neither 
the RPG program being registered (EMPDEPUDTF) nor any program that the registered 
RPG program calls will execute any SQL statements. 

 



Figure 8: External UDTF Syntax Diagram 
 
 

 
 
 
The PARAMETER STYLE clause is mandatory for external functions because it 

defines how DB2 will pass parameters to and from the registered program. The DB2SQL 
is the only style available for UDTFs not written in Java. We will explain this parameter 
style in detail in a moment. 

The SCRATCHPAD attribute is optional and provides the ability to define a memory 
area that can be shared across calls to the registered program. For example, this memory 
area could be used to store information from the last program call to be used as input for 
the next call to the external program by DB2. Maybe it’s a customized running total that 
cannot be done easily with SQL. This memory area defaults to a size of 100 bytes if an 
integer value is not specified. 

The NO FINAL CALL attribute starts to shed some light on how DB2 interacts with 
the registered program to return a result set to the invoker. The immediate reaction of 



most programmers is that they need to design a program that loops and populates the 
contents of the result set into a temporary data structure such as an array or user space. 
Actually, that is not the case at all. 

Instead, a single call of an UDTF results in DB2 invoking the underlying program 
multiple times. Figure 9 shows the different types of calls made by DB2 to the program. 
Because NO FINAL CALL has been specified for this UDTF, DB2 will invoke the 
registered RPG program a minimum of three times. This clause causes DB2 to bypass the 
First and Final Calls whenever the EmployeesDept table function is invoked. The 
external program is passed a parameter that contains the value of the call. This parameter 
enables the program to perform the action that DB2 is expecting to be performed for that 
call type. 

 
 
Figure 9: External UDTF Call Types 
 

 
 
 
To simplify this example, the program used the Open and Close call types to perform 

the setup and cleanup processing logically (associated with the First and Final call types, 
respectively). The First call type is designed to allow the external program to allocate 
resources that are one-time actions, such as getting ready the contents of the scratchpad 
for subsequent calls. This would be similar to the preparation and initialization of local 
host variables prior to opening an SQL cursor that references the variables. 

The Final call type can be used to perform the cleanup processing for the external 
program. As this example shows, this cleanup can also be done on the Close call type, but 
use of the Final call type may be useful when you’re trying to debug a possible bug on 
the Close call-type code in the external program. The Final Call type would allow one 
more call of the external program to do further analysis. 

 
 



  
In this example, an “Open” call type causes the a program to get things ready for 

returning a result set, which involves operations such as initializing the scratch pad area 
(if specified) and opening files. 

In a similar fashion, the “Close” call type would clean up the objects that were used 
during the all the program invocations. The “Fetch” call type is used by DB2 to call the 
external program multiple times until all of the rows in the result set have been returned. 

The program is invoked multiple times by DB2 with the Fetch call type to create the 
final result set, thus eliminating the need for the external program to loop and put the 
results into a temporary data structure. 

With a firm understanding of the key attributes for an external UDTF, a detailed 
walk through the underlying RPG program code is the next step. 
 
 
The RPG Code 

Starting with our F-specs in our RPG program (Figure 10), we include our S/36 file for 
the Department Master (DEP.MAST) and our Employee Master (EMP.MAST). Using 
the Extfile keyword, we can override the file names to program names to eliminate the 
period embedded in the file name, which often occurs with the S/36 naming convention. 
 
 Figure 10: RPG program F-specs 
  fDEPMAST   if   f   66     3aidisk    keyloc(1) extfile('DEP.MAST') 
  fEMPMAST   if   f   91     6aidisk    keyloc(1) extfile('EMP.MAST')  
 
 



The Prototype (PR) section (Figure 11) is where our definitions are mapped to match 
the parameter specifications required by the DB2SQL style. With this style, the input 
argument parameter (employee number) is passed first, and then all the output parameters 
returning the details for the specified employee are passed. The DB2SQL style requires 
all of the arguments to be null capable, so the next set of arguments are the null indicator 
variables for the input parameter and the indicator variables for all of the columns in the 
result set. 

 
Figure 11: RPG program prototype section  
    d EMPDEPUDTF         pr                  extpgm('EMPDEPUDTF ') 
 
     d arg_empno                      6a   varying 
 
     d res_empno                      6a   varying 
     d res_firstname                 14a   varying 
     d res_lastname                  17a   varying 
     d res_workdept                   3a   varying 
     d res_departname                38a   varying 
     d res_location                  16a   varying 
     d res_salary                     9p 2 
 
     d arg_empno_null... 
     d                                5i 0 
 
     d res_empno_null... 
     d                                5i 0 
     d res_firstname_null... 
     d                                5i 0 
     d res_lastname_null... 
     d                                5i 0 
     d res_workdept_null... 
     d                                5i 0 
     d res_departname_null... 
     d                                5i 0 
     d res_location_null... 
     d                                5i 0 
     d res_salary_null... 
     d                                5i 0 
 
 Following the null indicators is a set of variables that DB2 passes internally to the 

registered program (Figure 12). These arguments will never show up on the 
EmployeesDept table function invocation because the DB2 engine is responsible for 
them. 

Figure 12: RPG program internally passed variables 
     d  SQL_State                5 
     d  Function_Name          139    varying 
     d  Specific_Name          128    varying 
     d  Msg_Text                70    varying 
 
The SQL_State is an output parameter that is used to pass back the status of the table 

function processing to DB2. This program returns an SQLState value of '00000' each 
time a row of data is extracted successfully from the S/36 files. The value '02000' is 
returned when there is no more data to retrieve (i.e., end of file).  The next two 



parameters are input only, and they are not that interesting because DB2 just passes in the 
fully qualified name of the table function. The MSG_Text is an optional output parameter 
that can be used to pass descriptive text along with the SQLState value back to the 
invoker (Figure 13). 

 
Figure 13: ScratchPad parameter 
     d  ScratchPad                  10a   varying 
 
As we discussed earlier, the ScratchPad and Call Type parameters are much more 

interesting. The Scratchpad is the only memory that is persistent between calls to the 
EMPDEPUDTF program. This persistent memory can be used to store and pass values 
across program calls. The scratchpad used in this example is 10 bytes in size. The 
maximum size supported is 16 MB. The scratchpad will contain an 8-byte binary number 
containing the size of the scratchpad followed by storage of the specified size. 

We code a Scratch data structure (Figure 14) to map out the contents of the 
scratchpad memory area. In this example, the scratchpad area is used to store two values: 
an employee number and a flag to keep track of when the processing has been completed. 
The employee number field stores the next employee number value to process if all 
employees were selected (i.e., a blank employee number input argument). The next 
available employee is retrieved by reading ahead in the employee master table once the 
current employee has been processed. 

 
Figure 14: Sample scratchpad data structure 

d scratch         ds            10               inz 
d  sc_empno                      6a 
d  sc_done                       1a 

 
The done-processing flag signals to the program that end of table data has been 

reached and the SQL_State can be set to '02000'.  
The scratch pad usage in this example is for illustration purposes only. Keeping 

track of the next employee number to read is not necessary because the current position 
of the opened S/36 file will be retained between each call that DB2 makes to 
EMPDEPUDTF program. In addition, the processing flag is not needed because that 
event occurs on the DB2 “close” call to the UDTF program. A simplified version of the 
program that does not use the scratch pad is available at:  

http://ibm.com/iseries/db2/db2code.html 



The CallType integer (Figure 15) contains the call types of Open, Fetch, and Close, 
which we discussed earlier. Prototypes are defined (in Figure 16) for each of the call 
types that DB2 will be sending into the RPG code. This modular programming approach 
will allow for separate modules to be invoked for each call type value. 
 

Figure 15: The CallType integer 
     d  CallType                         5i 0 
Figure 16: Prototypes defined for each call type 
     d $close          pr 
     d $fetch          pr 
     d $open           pr 
     d getEMPMAST      pr 
     d setEndOfTable   pr 
 

Next, we need to specify a procedure interface that matches the prototype defined earlier 
for the DB2SQL parameter style (Figure 17). The parameter names and types match the 
previously defined PR values. 
 

Figure 17: Procedure interface that matches the prototype 
 

d EMPDEPUDTF      pi 
     d arg_empno                      6a   varying 
 
     d res_empno                      6a   varying 
     d res_firstname                 14a   varying 
     d res_lastname                  17a   varying 
     d res_workdept                   3a   varying 
     d res_departname                38a   varying 
     d res_location                  16a   varying 
     d res_salary                     9p 2 
 
     d arg_empno_null... 
     d                                5i 0 
     d res_empno_null... 
     d                                5i 0 
     d res_firstname_null... 
     d                                5i 0 
     d res_lastname_null... 
     d                                5i 0 
     d res_workdept_null... 
     d                                5i 0 
     d res_departname_null... 
     d                                5i 0 
     d res_location_null... 
     d                                5i 0 
     d res_salary_null... 
     d                                5i 0 
     d  SQL_State                     5 
     d  Function_Name               139    varying 
     d  Specific_Name               128    varying 
     d  Msg_Text                     70    varying 
     d  ScratchPad                   10a   varying 
     d  CallType                      5i 0 
 



Because we are processing S/36 files that are not externally defined, input 
specifications are coded to match the record layout of the department and employee 
master physical files (Figure 18). 

 
Figure 18: Input specs coded to match record layouts 

 
     iDEPMAST   ns 
     I                                  1    3  DEPTNO 
     I                                  6   41  DEPTNAME 
     I                                 42   47  MGRNO 
     I                                 48   50  ADMRDEPT 
     I                                 51   66  LOCATION 
     iEMPMAST   ns 
     i                                  1    6  EMPNO 
     i                                  9   20  FIRSTNAME 
     i                                 21   21  MIDINIT 
     i                                 24   38  LASTNAME 
     i                                 39   41  WORKDEPT 
     i                                 42   45  PHONENO 
     i                                 46   55  HIREDATE 
     i                                 56   63  JOB 
     i                             b   64   65 0EDLEVEL 
     i                                 66   66  SEX 
     i                                 67   76  BIRTHDATE 
     i                             p   77   81 2SALARY 
     i                             p   82   86 2BONUS 
     i                             p   87   91 2COMM 
 
Compared to the complicated parameter list required by DB2, the mainline 

processing section of our RPG program is relatively simple (Figure 19). The mainline 
logic revolves around the Call Type being passed in and the routing of the call to the 
appropriate procedures. 

 
Figure 19: RPG program mainline processing section 

 
select; 
       when CallType = -1; 
        callp $open(); 
       when CallType = *zero; 
        callp $fetch(); 
       when CallType = 1; 
        callp $close(); 
endsl; 
 
return; 
 



The $OPEN procedure (Figure 20) is performed during the first call of the RPG 
program by the DB2 engine. DB2 initializes the scratchpad with binary zeros before 
calling the program for the first time, and DB2 also sends the address of this area to the 
RPG program for each type of call. 

 
Figure 20: The $OPEN procedure 
     p $open           b 
 
     d $open           pi 
 
      /free 
 
       clear scratch; 
 
       ScratchPad = scratch; 
 
       return; 
 
      /end-free 
 
     p $open           e 
 
The $FETCH procedure (Figure 21) is where the extraction of data from our S/36 

files takes place. This procedure controls the retrieval of data from the employee master 
file and the related department file. 

 
Figure 21: The $FETCH procedure 
     p $fetch          b 
 
     d $fetch          pi 
 
      /free 
 
       scratch = ScratchPad; 
 
       get_empno = sc_empno; 
 
       if sc_done = 'Y'; 
        callp setEndOfTable(); 
        return; 
       endif; 
 
       callp getEMPMAST(); 
 
       ScratchPad = scratch; 
 
       return; 
 
      /end-free 
 
     p $fetch          e 
 
The first step in the $FETCH procedure is having the local Scratch data structure 

reset with the scratch pad parameter data to map out the location of the work fields. 



The procedure checks to see if the previous call marked the data retrieval as 
complete by setting the done flag (SC_DONE). If all of the data has been returned, the 
end-of-file condition must be returned to DB2 with the SQL_STATE parameter; this is 
performed in the setEndOfTable procedure call.  If there is still data to return, the 
getEMPMAST procedure is called using the next employee number to retrieve from the 
scratch pad area. 

The $CLOSE procedure (Figure 22) is invoked once all of the data has been 
returned. This procedure signals that the end-of-file condition has been reached by 
invoking the SetEndOfTable procedure. 

 
Figure 22: The $CLOSE procedure 
p $close          b 
 
     d $close          pi 
 
      /free 
 
       callp setEndOfTable(); 
 
       return; 
 
      /end-free 
 
     p $close          e 
 

The getEMPMAST procedure (Figure 23) is where the records are physically retrieved.  
If the invoker of the UDTF supplies a valid employee identifier number (at A in Figure 
23), then the program returns the personnel and department data for that specified 
employee. If the employee is not found, the setEndOfTable procedure is called to set the 
SQL_STATE to '02000' and the returned parameters to NULL to indicate that no data 
was found, and control returns to the caller. 

Figure 23: The getEMPMAST procedure 
     p getEMPMAST      b 
 
     d getEMPMAST      pi 
 
      /free 
 
       select; 
 
    A  when arg_empno <> *blanks 
 
         get_empno = arg_empno; 
 
         chain(e) get_empno EMPMAST  ; 
 
         if not %found(EMPMAST); 
           res_empno = *blanks; 
           callp setEndOfTable(); 
           return; 
         endif; 
 
         sc_done = 'Y'; 

when arg_empno <> *blanks 



         res_empno     = empno; 
         res_firstname  = firstname; 
         res_lastname  = lastname; 
         res_workdept  = workdept; 
         res_salary    = salary; 
 
         chain(e) workdept DEPMAST; 
 
         if not %found(DEPMAST); 
           res_departname= *blanks; 
           res_location  = *blanks; 
         else; 
           res_departname = deptname; 
           res_location = location; 
         endif; 

 
        B       when arg_empno = *blanks 

 
         setll get_empno EMPMAST; 
 
         read(e) EMPMAST; 
 
         if %eof(EMPMAST); 
           res_empno = *blanks; 
           clear sc_empno; 
           sc_done = 'Y'; 
           callp setEndOfTable(); 
           return; 
         endif; 
 
         res_empno     = empno; 
         res_firstname  = firstname; 
         res_lastname  = lastname; 
         res_workdept  = workdept; 
         res_salary    = salary; 
 
         chain(e) workdept DEPMAST  ; 
 
         if not %found(DEPMAST); 
           res_departname= *blanks; 
           res_location  = *blanks; 
         else; 
           res_departname = deptname; 
           res_location = location; 
         endif; 
 
         read EMPMAST; 
 
         if not %eof(EMPMAST); 
           sc_empno = empno; 
           clear sc_done; 
         else; 
           sc_done = 'Y'; 
         endif; 
       endsl; 
 
       return; 
      /end-free 
     p getEMPMAST      e 

when arg_empno = *blanks 



 
The scratchpad field SC_DONE is set to Y to indicate on the next DB2 fetch call that 

all data has been retrieved for the specified employee, as was previously mentioned in the 
$FETCH procedure above. 

When a record of the S/36 file is identified to be retrieved, the data for each 
programmed-described field is copied to the result output parameters (e.g., res_empno, 
res_firstname). All of the result null indicators are initialized to not null because S/36 
files do not contain any null values. This processing is what allows DB2 to pass the result 
data back to the invoker and make it appear like data from a newly created SQL table 
instead of an old S/36 file. 

If the invoker of the UDTF supplies a blank employee identifier (at B in Figure 23), 
then the procedure will execute logic to return the data for all of the employees found in 
the employee and department files. In this case, the SC_EMPNO field in the Scratchpad 
area is used to keep the next employee number to be processed, so that the next fetch call 
type from DB2 will result in the program resuming processing on the next employee. 

The SetEndOfTable procedure (Figure 24) is used by the program to flag end-of-file 
by setting the SQL_State variable to '02000'. This procedure is called to end processing 
when data for either a single employee or all employees is returned by the table function. 

 
Figure 24: The SetEndOfTable procedure 
 
p setEndOfTable   b 
 
     d setEndOfTable   pi 
 
      /free 
 
       SQL_State = '02000'; 
 
       clear res_empno     ; 
       clear res_firstname  ; 
       clear res_lastname  ; 
       clear res_workdept  ; 
       clear res_departname; 
       clear res_location  ; 
       clear res_salary    ; 
 
       res_empno_null       = -1; 
       res_firstname_null   = -1; 
       res_lastname_null    = -1; 
       res_workdept_null    = -1; 
       res_departname_null  = -1; 
       res_location_null    = -1; 
       res_salary_null      = -1; 
 
       Msg_Text = 'End of Table'; 
 
       return; 
 
       /end-free 
 
     p setEndOfTable   e  

 



The final piece of the example is demonstrating how to call the EmployeesDept table 
function (Figure 25). Compared to the earlier examples, it should be clear that the invoker 
will not know whether he or she is receiving data from a DB2 table or data extracted 
from a legacy S/36 file. This SELECT statement will retrieve the specified columns from 
the S/36 files for the employee with an identifier value of '000030'. 

 
Figure 25: Calling the EmployeesDept table function 
 
SELECT empno, firstname, lastname, deptname  
    FROM TABLE(employeesdept('000030')) myudtf2 
 

 
UDTF Coding Tips 

The secondary threads used by DB2 for calling UDFTs can hold resources such as locks 
that can hinder concurrent system activity on shared objects. Due to this fact, IBM 
recommends that UDTFs perform short, quick-running requests. By default, UDFs will 
time out and end execution after 30 seconds. If you need to allow a UDTF to run for a 
longer amount of time, you can adjust the timeout value by specifying the 
UDF_TIME_OUT attribute in a QAQQINI query options file. 

On the topic of lock conflicts, it’s also not a good idea to have the UDTF execute 
operations on the same database objects that are referenced by the SQL statement that 
invoked the UDTF in the first place. 

The use of threads by DB2 also makes it trickier to implement UDFs with the non-
ILE languages, which are not thread-safe. You can use the non-ILE languages for 
UDTFs, but we highly recommend you use the ILE languages due to their support of 
threads. 

With the examples we’ve reviewed, you now have the basic foundation for the 
design and coding of both SQL and external user-defined table functions. We hope we’ve 
provided you a deeper appreciation of the nonrelational data access and integration 
problems that can be solved with UDTFs. 

